

DEX ... PLOT

PURPOSE

Generates a dex plot for a given statistic.

DESCRIPTION

A dex statistic plot is a graphical method for representing a design of experiment problem. The first variable is a response variable while the remaining variables (must be at least one) represent levels of factors. The user specified statistic is computed for each level of each factor. A separate subplot is drawn for each factor with the subplot for factor k centered horizontally at $x=k$. Each subplot has a given horizontal width (defined by the DEX WIDTH command, defaults to 0.5). For example, the subplot for factor 2 ranges from 1.75 to 2.25 on the horizontal axis. The levels of the factor are assigned an x coordinate within this range (from lowest to highest). Then within each subplot:

Vertical axis = value of the computed statistic from the response variable for a given level of the factor;
Horizontal axis = value of the level of a given factor.

This plot graphically shows the following:

1. How the statistic for the response variable varies with the level of the factor;
2. How the statistic for the response variable varies between factors.

SYNTAX 1

DEX <stat> PLOT <y> <x1> ... <xn> <SUBSET/EXCEPT/FOR qualification>

where <y> is the response variable;

<x1> ... <xn> is a sequence of variables representing factors in a designed experiment;

<stat> is one of the following statistics:

MEAN (or AVERAGE), MIDMEAN, MEDIAN, TRIMMED MEAN, WINDSORIZED MEAN,
SUM, PRODUCT, SIZE (or NUMBER or COUNT), MINIMUM, MAXIMUM,
STANDARD DEVIATION (or SD), VARIANCE,
STANDARD DEVIATION OF MEAN (or SDM), VARIANCE OF MEAN (or VM),
RELATIVE STANDARD DEVIATION (or RELSD),
RELATIVE VARIANCE (or RELV or COEFFICIENT OF VARIATION),
RANGE, MIDRANGE, LOWER HINGE, UPPER HINGE, LOWER QUARTILE, UPPER QUARTILE,
<FIRST/SECOND/THIRD/FOURTH/FIFTH/SIXTH/SEVENTH/EIGHTH/NINTH> DECILE (or 1DEC, 2DEC,
3DEC,4DEC,5DEC,6DEC,7DEC, 8DEC, 9DEC),
SKEWNESS, KURTOSIS, PROPORTION,
AUTOCORRELATION, AUTOCOVARIANCE,
SINE FREQUENCY, SINE AMPLITUDE,
CP, CPK, EXPECTED LOSS, PERCENT DEFECTIVE,
SINE FREQUENCY, SINE AMPLITUDE,
TAGUCHI SN, TAGUCHI SN+, TAGUCHI SN-, TAGUCHI SN00;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for statistics that only require a single variable to compute.

SYNTAX 2

DEX <stat> PLOT <y> <x> <x1> ... <xn> <SUBSET/EXCEPT/FOR qualification>

where <y> is the response variable;

<x> is a second variable used in calculating the statistic (e.g., a linear fit is computed between <y1> and <x>);

<x1> ... <xn> are a sequence of variables representing factors in a designed experiment;

<stat> is one of the following statistics:

LINEAR INTERCEPT, LINEAR SLOPE, LINEAR RESSD, LINEAR CORRELATION,

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for statistics that require two variables to compute.

EXAMPLES

```
DEX MEAN PLOT Y X1 X2
DEX MEDIAN PLOT Y X1 X2
DEX SD PLOT Y X1 X2 X3
```

DEX RANGE PLOT Y X1 X2 X3 X4
 DEX SD PLOT Y X1 TO X3
 DEX RANGE PLOT Y X1 TO X4

NOTE 1

This plot is normally done for a location parameter (typically the mean or median) or a spread parameter (typically the standard deviation or range). The other statistics are less often used.

NOTE 2

The TO syntax is allowed for the list of factor variables (see the EXAMPLES above).

NOTE 3

The CHARACTER and LINE settings can be used to control the appearance of the plot. The first trace is typically drawn with a blank line and some type of character set (the choice of character is a matter of user preference). The second trace draws a horizontal line at the value for the specified statistic for the entire response variable. This is typically drawn with a blank character and a solid line (some analysts may prefer a dashed or dotted line). In any event, the user must explicitly set character and line settings (they default to all lines solid and all characters blank).

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

LINES	=	Sets the type for plot lines.
CHARACTER	=	Sets the type for plot characters
PLOT	=	Generates a data or function plot.
DEX SCATTER PLOT	=	Generates a dex scatter plot.
DEX SIGN PLOT	=	Generates a dex sign plot.
DEX ... PARETO PLOT	=	Generates a Pareto dex plot for a statistic.
DEX ... YOUNDEN PLOT	=	Generates a Youden dex plot for a statistic.
DEX ... EFFECTS PLOT	=	Generates a dex effects plot for a statistic.
DEX ... PARETO EFFECTS PLOT	=	Generates a Pareto effects dex plot for a statistic.
DEX ... ABSOLUTE EFFECTS PLOT	=	Generates an absolute effects dex plot for a statistic.
DEX ... PARE ABSO EFFECTS PLOT	=	Generates a Pareto absolute effects dex plot for a statistic.
DEX WIDTH	=	Specifies the width of levels in a dex plot.

REFERENCE

“Statistics for Experimenters,” Box, Hunter, and Hunter, Wiley and Sons, 1978.

APPLICATIONS

Design of Experiments

IMPLEMENTATION DATE

89/12

PROGRAM

```

SKIP 25
READ SHEESLE2.DAT Y PROC PLANT SPEED SHIFT PROC
.
CHARACTERS X BLANK
LINES SOLID SOLID
LET NFACT = 4
XLIMITS 1 NFACT
MAJOR XTIC MARK NUMBER NFACT
MINOR XTIC MARK NUMBER 0
XTIC OFFSET 1 1
XTIC LABEL FORMAT ALPHA
XTIC LABEL CONTENT PROCESS PLANT SPEED SHIFT
X1LABEL FACTORS
YLIMITS 20 28
.
MULTIPLY 2 2
MULTIPLY CORNER COORDINATES 0 0 100 100
Y1LABEL MEAN
DEX MEAN PLOT Y PROC PLANT SPEED SHIFT
Y1LABEL MEDIAN
DEX MEDIAN PLOT Y PROC PLANT SPEED SHIFT
YLIMITS
Y1LABEL STANDARD DEVIATION
DEX SD PLOT Y PROC PLANT SPEED SHIFT
Y1LABEL RANGE
DEX RANGE PLOT Y PROC PLANT SPEED SHIFT
END OF MULTIPLY

```

